

CAPITOLUL 3

ELEMENTELE DE BAZA ALE LIMBAJULUI DE ASAMBLARE

Limbajul mașină al unui sistem de calcul (SC) este format din totalitatea instrucțiunilor mașină puse la dispoziție de procesorul SC. Acestea se reprezintă sub forma unor șiruri de biți cu semnificație prestabilită.

Limbajul de asamblare al unui calculator este un limbaj de programare în care setul de bază al instrucțiunilor coincide cu operațiile mașinii și ale cărui structuri de date coincid cu structurile primare de date ale mașinii. **Limbaj simbolic. Simboluri - Mnemonice + etichete.**

Elementele cu care lucrează un **asamblor** sunt:

- * **etichete** - nume scrise de utilizator, cu ajutorul cărora se pot referi date sau zone de memorie.
- * **instrucțiuni** - scrise sub forma unor mnemonice care sugerează acțiunea. Asamblorul generează octeții care codifică instrucțiunea respectivă.
- * **directive** - sunt indicații date asamblorului în scopul generării corecte a octetilor. Ex: relații între modulele obiect, definirea unor segmente, indicații de asamblare condiționată, directive de generare a datelor.
- * **contor de locații** - număr întreg gestionat de asamblor. În fiecare moment, valoarea contorului coincide cu numărul de octeți generați corespunzător instrucțiunilor și directivelor deja întâlnite în cadrul segmentului respectiv (deplasamentul curent în cadrul segmentului). Programatorul poate utiliza această valoare (accesare doar în citire!) prin simbolul '\$'. **Fiecare segment are propriul său contor de locații !!!**

NASM supports two special tokens in expressions, allowing calculations to involve the current assembly position: the \$ and \$\$ tokens. \$ evaluates to the assembly position at the beginning of the line containing the expression; so you can code an infinite loop using `JMP $`.

\$\$ evaluates to the start of the current section; so you can tell how far into the section are by using `($-$)`.

Directiva SECTION

```
section .data
    db 'hello'
    db 'h', 'e', 'l', 'l', 'o'
    data_segment_size equ $-$$
```

\$-\$\$ = the distance from the beginning of the segment AS A SCALAR (constant numerical value)!!!!!!!!!!

\$ - is an offset = POINTER TYPE !!!! It is an address !!!!

\$\$ - is an offset =POINTER TYPE !!!! It is an address !!!!

\$ means "address of here".

\$\$ means "address of start of current section".

So \$-\$\$ means "current size of section".

For the example above, this will be 10, as there are 10 bytes of data given.

Daca nu am definite sectiuni explicite , atunci \$\$=inceputul segmentului respectiv

3.1. FORMATUL UNEI LINII SURSĂ

Formatul unei linii sursă în limbajul de asamblare x86 este următorul:

[*etichetă*[:]] [*prefixe*] [*mnemonică*] [*operandi*] [*;comentariu*]

Ilustrăm conceptul prin intermediul a câteva exemple de linii sursă:

aici: jmp acolo	; avem etichetă + mnemonică + operand + comentariu
repz cmpsd	; prefix + mnemonică + comentariu

start: ; etichetă + comentariu
 ; doar un comentariu (care putea lipsi si el)
a dw 19872, 42h ; etichetă + mnemonică + 2 operanzi + comentariu

Caracterele din care poate fi constituită o *etichetă* sunt următoarele:

- Litere, atât A-Z cât și a-z;
- Cifre de la 0 la 9;
- Caracterele _, \$, \$\$, #, @, ~, . și ?

Ca și prim caracter al unei etichete sunt permise doar litere, _ și ?

Aceste reguli sunt valabile pentru toți *identificatorii* valizi (denumiri simbolice, precum nume de variabile, etichete, macro, etc).

Identificatorii NASM sunt *case sensitive*, limbajul diferențiind literele mari de cele mici privitor la denumirile utilizator. Aceasta înseamnă că un identificator Abc este diferit de identificatorul abc. Pentru denumirile implicit parte a limbajului, cum ar fi cuvintele cheie, mnemonicile și numele regiștrilor, nu se diferențiază literele mari de cele mici (acestea sunt case insensitive).

La nivelul limbajului de asamblare se întâlnesc două categorii de etichete:

1). ***etichete de cod***, care apar în cadrul secvențelor de instrucțiuni cu scopul de a defini destinațiile de transfer ale controlului în cadrul unui program. **Pot apărea și în segmente de date !**

2). ***etichete de date***, care identifică simbolic unele locații de memorie, din punct de vedere semantic ele fiind echivalentul noțiunii de *variabilă* din alte limbaje. **Pot apărea și în segmente de cod !**

Valoarea unei etichete în limbaj de asamblare este un număr întreg reprezentând adresa instrucțiunii, directivei sau datelor ce urmează etichetei.

Distincția dintre referirea adresei unei variabile sau a conținutului asociat acesteia în NASM se face după regulile:

- Când este specificat **între paranteze drepte, numele variabilei desemnează valoarea variabilei**, de exemplu [p] specifică accesarea valorii variabilei p, similar cu modul în care *p semnifică dereferențierea unui pointer (accesul la conținutul indicat prin valoarea pointerului) în C;
- În orice alt context **numele variabilei reprezintă adresa variabilei**, spre exemplu, p este întotdeauna adresa variabilei p;

Exemple:

mov EAX, et ; încarcă în registrul EAX **adresa** datelor sau a codului marcat cu eticheta et (4 octeți)
mov EAX, [et] ; încarcă în registrul EAX **conținutul** de la adresa et (4 octeți)
lea eax, [v] ; încarcă în registrul eax **adresa** (offsetul) variabilei v (4 octeți)

Ca generalizare, **folosirea parantezelor pătrate indică întotdeauna accesarea unui operand din memorie**. De exemplu, mov EAX, [EBX] semnifică un transfer în EAX a conținutului memoriei a cărei adresă este dată de valoarea lui EBX.

Există două tipuri de *mnemonice*: mnemonice de *instrucțiuni* și nume de *directive*. Directivele dirijează asamblorul. Ele specifică modul în care asamblorul va genera codul obiect. Instrucțiunile dirijează procesorul.

Operanzii sunt parametri care definesc valorile ce vor fi prelucrate de instrucțiuni sau de directive. Ei pot fi **registri, constante, etichete, expresii, cuvinte cheie sau alte simboluri**. Semnificația operanzilor depinde de mnemonica instrucțiunii sau directivei asociate.

3.2. EXPRESII

expresie - operanzi + operatori. *Operatorii* indică modul de combinare a operanzilor în scopul formării expresiei. **Expresiile sunt evaluate în momentul asamblării** (adică, valorile lor sunt determinabile la momentul asamblării, cu excepția acelor părți care desemnează conținuturi de regiștri și care vor fi determinate la execuție).

3.2.1. Moduri de adresare

Operanzii instrucțiunilor pot fi specificați sub forme numite *moduri de adresare*. Cele trei tipuri de operanzi sunt: *operanzi imediați*, *operanzi registru* și *operanzi în memorie*.

Valoarea operanzilor este calculată în momentul asamblării pentru operanzii imediați și pentru offset-urile reprezentând adresarea directă, în momentul încărcării programului pentru adresarea directă (**adresa FAR**) și în momentul execuției pentru operanzii registru și cei adresați indirect.

?:offset (assembly time) 0708:offset (loading time)

3.2.1.1. Utilizarea operanzilor imediați

Operanzii imediați sunt formați din date numerice constante calculabile la momentul asamblării.

Constantele întregi se specifică prin valori binare, octale, zecimale sau hexazecimale. Adicional, este permisă folosirea caracterului _ (underscore) pentru a separa grupuri de cifre. Baza de numerație poate fi precizată în mai multe moduri:

- Folosind sufixele H sau X pentru hexazecimal, D sau T pentru zecimal, Q sau O pentru octal și B sau Y pentru binar; în aceste cazuri numărul trebuie să înceapă obligatoriu cu o cifră între 0 și 9, pentru a nu exista confuzii între constante și simboluri, de exemplu, OABCH este interpretat ca număr hexazecimal, dar ABCH este interpretat ca simbol
- În stil C, prin prefixare cu 0x sau 0h pentru hexazecimal, 0d sau 0t pentru zecimal, 0o sau 0q pentru octal, respectiv 0b sau 0y pentru binar;

Exemple:

- constanta hexazecimală B2A poate fi exprimată ca 0xb2a, 0xb2A, 0hb2a, 0b12Ah, 0B12AH, etc;
- valoarea zecimală 123 poate fi specificată ca 123, 0d123, 0d0123, 123d, 123D, ...
- 11001000b, 0b11001000, 0y1100_1000, 001100_1000Y reprezintă diferite exprimări ale nr bin 11001000

Deplasamentele etichetelor de date și de cod reprezintă valori determinabile la momentul asamblării care rămân constante pe tot parcursul execuției programului

mov eax, et ; transfer în registrul EAX a adresei (offsetului) asociate etichetei et

va putea fi evaluată la momentul asamblării drept de exemplu

mov eax, 8 ; distanță de 8 octeți față de începutul segmentului de date

“Constanța” acestor valori derivă din regulile de alocare adoptate de limbajele de programare în general și care statuează că ordinea de alocare în memorie a variabilelor declarate (mai precis distanța față de începutul segmentului de date în care o variabilă este alocată) sau respectiv distanțele salturilor destinație în cazul unor instrucțiuni de tip **goto** sunt valori constante pe parcursul execuției unui program.

Adică: o variabilă odată alocată în cadrul unui segment de memorie nu își va schimba niciodată locul alocării (adică poziția sa față de începutul acelui segment) iar această informație determinabilă la momentul asamblării derivă din ordinea specificării variabilelor la declarare în cadrul textului sursă și din dimensiunea de reprezentare dedusă pe bază informației de tip asociate.

3.2.1.2. Utilizarea operanzilor registru

Modul de *invocare/accesare directă* - **mov eax, ebx**

Invocare/accesare *indirectă* - pentru a indica locațiile de memorie - **mov eax, [ebx]**

3.2.1.3. Utilizarea operanzilor din memorie

Operanzii din memorie : cu *adresare directă* și cu *adresare indirectă*.

Operandul cu *adresare directă* este o constantă sau un simbol care reprezintă adresa (segment și deplasament) unei instrucțiuni sau a unor date. Acești operanzi pot fi *etichete* (de ex: jmp et, var1 dw 324, add eax,[b]), *nume de proceduri* (de ex: call proc1) sau *valoarea contorului de locații* (de ex: b db \$-a).

Deplasamentul unui operand cu adresare directă este calculat în momentul asamblării (assembly time). Adresa fiecărui operand raportată la structura programului executabil (mai precis stabilirea segmentelor la care se raportează deplasamentele calculate) este calculată în momentul editării de legături (linking time). Adresa fizică efectivă este calculată în momentul încărcării programului pentru execuție (loading time – acest proces final de ajustare a adreselor numindu-se RELOCAREA ADRESELOR = Address Relocation).

Un deplasament utilizat ca operand în cadrul unui program este întotdeauna raportat la un registru de segment. Acest registru poate fi specificat explicit sau, în caz contrar, se asociază de către asamblor în mod implicit un registru de segment. Regulele limbajului de asamblare pentru asocierile implicite sunt:

- **CS** pentru etichete de cod destinație ale unor salturi (jmp, call, ret, jz etc);
- **SS** în adresări SIB ce folosește EBP sau ESP drept bază (indiferent de index sau scală);
- **DS** pentru restul accesărilor de date

Specificarea explicită a unui registru de segment se face cu ajutorul operatorului de prefixare segment (notat ":" și care se mai numește, 'operatorul de specificare a segmentului'). ES poate fi utilizat numai în specificări explicite (ca de exemplu ES:[Var] sau ES:[ebx+eax*2-a]) sau în cadrul unor instrucțiuni pe siruri (MOVSB)

JMP FAR CS:...

JMP FAR DS:... or JMP FAR [label2]

3.2.1.4. Operanzi cu adresare indirectă

Operanzii cu *adresare indirectă* utilizează regiștri pentru a indica adrese din memorie. Deoarece valorile din regiștri se pot modifica la momentul execuției, adresarea indirectă este indicată pentru a opera în mod dinamic asupra datelor.

Forma generală pentru accesarea indirectă a unui operand de memorie este dată de formula de calcul a offset-ului unui operand:

$$[\text{registru_de_bază}] + [\text{registru_index} * \text{scală}] + [\text{constantă}]$$

Constanta este o expresie a cărei valoare este determinabilă la momentul asamblării. De exemplu, $[\text{ebx} + \text{edi} + \text{table} + 6]$ desemnează un operand prin adresare indirectă, unde atât *table* cât și 6 sunt constante.

Operanzii *registru_de_bază* și *registru_index* sunt folosiți de obicei pentru a indica o adresă de memorie referitoare la un tablou. În combinație cu factorul de scalare, mecanismul este suficient de flexibil pentru a permite acces direct la elementele unui tablou de înregistrări, cu condiția ca dimensiunea în octeți a unei înregistrări să fie 1, 2, 4 sau 8. De exemplu, octetul superior al elementului de tip DWORD cu index dat în ecx, parte a unui vector de înregistrări al cărui adresă (a vectorului) este în edx poate fi încărcat în dh prin intermediul instrucțiunii

```
mov dh, [edx + ecx * 4 + 3]
```

Din punct de vedere sintactic, atunci când operandul nu este specificat prin formula completă, lipsind unele dintre componente (de exemplu lipsește “* scală”), asamblorul va rezolva ambiguitatea care rezultă printr-un proces de analiză a tuturor formele echivalente de codificare posibile și alegerea celei mai scurte dintre acestea. Cu alte cuvinte, având

```
push dword [eax + ebx]           ; salvează pe stivă dublucuvântul de la adresa eax+ebx
```


asamblorul are libertatea de a considera eax drept bază și ebx drept index sau invers, ebx drept bază și eax drept index. Analog, pentru

pop DWORD [ecx] ; restaurează vârful stivei în variabila cu adresa dată de ecx

asamblorul poate interpreta ecx fie ca bază fie ca index. Ce este realmente important de reținut este faptul că toate codificările luate în considerare de către asamblor sunt echivalente iar decizia finală a asamblorului nu are impact asupra funcționalității codului rezultat.

De asemenea, în plus față de rezolvarea unor astfel de ambiguități, asamblorul permite și exprimări non-standard cu condiția ca acestea să fie transformabile într-un final în forma standard de mai sus.

Alte exemple:

lea eax, [eax*2] ; încarcă în eax valoarea lui eax*2 (adică, eax devine 2*eax)

În acest caz, asamblorul poate decide între codificare de tip bază = $eax + index = eax$ și scală = 1 sau index = eax și scală = 2.

lea eax, [eax*9 + 12] ; eax ia valoarea $eax * 9 + 12$

Deși scală nu poate fi 9, asamblorul nu va emite aici un mesaj de eroare. Aceasta deoarece el va observa posibila codificare a adresei drept: bază= $eax + index = eax$ cu scală=8, unde de această dată valoarea 8 este corectă pentru scală. Evident, instrucțiunea putea fi precizată mai clar sub forma

lea eax, [eax + eax * 8 + 12].

Să reținem deci că pentru adresarea indirectă, esențială este specificarea între paranteze drepte a cel puțin unuia dintre elementele componente ale formulei de calcul a offsetului.

3.2.2. Utilizarea operatorilor

Operatori - pentru combinarea, compararea, modificarea și analiza operanzilor. Unii operatori lucrează cu constante întregi, alții cu valori întregi memorate, iar alții cu ambele tipuri de operanzi.

Este importantă înțelegerea diferenței dintre operatori și instrucțiuni. **Operatorii efectuează calcule cu valori constante SCALARE determinabile la momentul asamblării** (valori scalare = valori imediate), cu excepția adunării și scaderii unei constante la un/dintr-un pointer (care va furniza întotdeauna “pointer data type”) și cu excepția formulei de calcul al offset-ului unui operand (care permite operatorul ‘+’). Instrucțiunile efectuează calcule cu valori ce pot fi necunoscute până în momentul execuției. Operatorul de adunare (+) efectuează adunarea în momentul asamblării; instrucțiunea ADD efectuează adunarea în timpul execuției.

Operatorii disponibili pentru construcția expresiilor sunt asemănători celor din limbajul C, atât ca sintaxă cât și din punct de vedere semantic. **Evaluarea expresiilor numerice se face pe 64 de biți**, rezultatele finale fiind ulterior ajustate în conformitate cu dimensiunea de reprezentare disponibilă în contextul de utilizare al expresiei.

În tabelul de mai jos sunt prezentați în ordinea priorității operatorii ce pot fi folosiți în cadrul expresiilor limbajului de asamblare x86 (NASM !!).

Prioritate	Operator	Tip	Rezultat
7	-	Unar, prefixat	Complement față de 2 (negare): $-X = 0 - X$
7	+	Unar, prefixat	Fără efect (oferit pentru simetrie cu „-”): $+X = X$
7	~	Unar, prefixat	Complement față de 1: <code>mov al, ~0 => mov AL, 0xFF</code>
7	!	Unar, prefixat	Negare logică: $!X = 0$ când $X \neq 0$, altfel 1
6	*	Binar, infix	Înmulțire: $1 * 2 * 3 = 6$
6	/	Binar, infix	Câtul împărțirii fără semn: $24 / 4 / 2 = 3$ ($-24 / 4 / 2 = 0FDh$)
6	//	Binar, infix	Câtul împărțirii cu semn: $-24 // 4 // 2 = -3$ ($-24 / 4 / 2 \neq -3!$)
6	%	Binar, infix	Restul împărțirii fără semn: $123 \% 100 \% 5 = 3$
6	%%	Binar, infix	Restul împărțirii cu semn: $-123 \% \% 100 \% \% 5 = -3$
5	+	Binar, infix	Însumare: $1 + 2 = 3$
5	-	Binar, infix	Scădere: $1 - 2 = -1$

4	<<	Binar, infix	Deplasare pe biți către stânga: $1 \ll 4 = 16$
4	>>	Binar, infix	Deplasare pe biți la dreapta: $0xFE \gg 4 = 0x0F$
3	&	Binar, infix	ȘI: $0xF00F \& 0xFF6 = 0x0006$
2	^	Binar, infix	SAU exclusiv: $0xFF0F \wedge 0xF0FF = 0xFF0$
1		Binar, infix	SAU: $1 2 = 3$

Operatorul de indexare are o utilizare largă în specificarea operanzilor din memorie adresați indirect. Paragraful 3.2.1 a clarificat rolul operatorului [] în adresarea indirectă.

Acest tabel este de o importanță covârșitoare în anumite situații și e bine să îl avem „la îndemână”. Iată de ce:

$5|6+7\&8 = (5|6)+(7\&8) = 7+0 = 7$?? NU !!!!!!! datorită precedenței operatorilor avem;

$5|6+7\&8 = 5|(6+7)\&8 = 5|13\&8 = 5|8=13 = 0Dh$!!!

3.2.2.3. Operatori de deplasare de biți

expresie >> cu_cât și *expresie << cu_cât*

mov ah, 01110111b << 3 ; desemnează valoarea 10111000b

add bh, 01110111b >> 3 ; desemnează valoarea 00001110b

AX 00000011 10111000 !!!!!

BX 00000000 00001110b

3.2.2.4. Operatori logici pe biți

Operatorii pe biți efectuează operații logice la nivelul fiecărui bit al operandului (operanzilor) unei expresii. Expresiile au ca rezultat valori constante.

OPERATOR	SINTAXA	SEMNIFICAȚIE
~	~ expresie	complementare biți
&	expr1 & expr2	ȘI bit cu bit
	expr1 expr2	SAU bit cu bit
^	expr1 ^ expr2	SAU exclusiv bit cu bit

Exemple (presupunem că expresia se reprezintă pe un octet):

~ 11110000b ; desemnează valoarea 00001111b ; ~0f0h = ~...
01010101b & 11110000b ; are ca rezultat valoarea 01010000b
01010101b | 11110000b ; are ca rezultat valoarea 11110101b
01010101b ^ 11110000b ; are ca rezultat valoarea 10100101b
! – negare logică (similar cu limbajul C) ; !0 = 1 ; !(orice diferit de zero) = 0

3.2.2.6. Operatorul de specificare a segmentului

Operatorul de specificare a segmentului (:) comandă calcularea adresei FAR a unei variabile sau etichete în funcție de un anumit segment. Sintaxa este: *segment:expresie*

[ss: ebx+4]; deplasamentul e relativ la SS // [es:082h] ; deplasamentul e relativ la ES
10h:var ; segmentul este indicat de selectorul 10h, iar offsetul este valoarea etichetei *var*.

3.2.2.7. Operatori de tip

Specifică tipurile unor expresii și a unor operanzi păstrați în memorie. Sintaxa pentru aceștia este

tip expresie

unde specificatorul de tip este unul dintre cuvintele cheie **BYTE**, **WORD**, **DWORD**, **QWORD**.

Această construcție sintactică forțează ca *expresie* să fie tratată ca având dimensiunea de reprezentare *tip*, fără însă a-i modifica definitiv (distructiv) valoarea în sensul precizat de conversia dorită. De aceea, aceștia sunt considerați **operatori de conversie (temporară) nedistructivă**. Pentru operanzii păstrați în memorie, *tip* poate fi **BYTE**, **WORD**, **DWORD**, **QWORD** având dimensiunile de reprezentare 1, 2, 4, 8 octeți. Pentru etichetele de cod el poate fi **NEAR** (adresă pe 4 octeți) sau **FAR** (adresă pe 6 octeți). Expresia **byte** [A] va indica doar primul octet de la adresa indicată de A. Analog, **dword** [A] indică dublucuvântul ce începe la adresa A.

Specificatorii BYTE / WORD / DWORD / QWORD au intotdeauna doar rol de a clarifica o ambiguitate (inclusiv cand este vorba despre o variabila de memorie, faptul de a preciza **mov BYTE [v], 0** sau **mov WORD [v], 0** este tot o clarificare a ambiguitatii, cum nasm nu asociaza faptul ca v este byte/ word / dword).

mov [v],0 ; syntax error – operation size not specified

Specificatorul QWORD nu intervine niciodată explicit în cod pe 32 de biți.

Exemple unde e necesar un specificator de dimensiune al operanzilor:

- mov [mem], 12 - (i)div [mem] ; (i)mul [mem]

- push [mem] ; pop [mem]

- **push 15** - aici este o inconsistență în NASM, asamblorul nu va emite eroare/warning ci va face
- **push DWORD 15**

Exemple de operanzi IMPLICITI efectiv pe 64 biți (în cod pe 32):

- **mul dword [v]** ; înmulțește eax cu dword-ul de la adresa v și depune în EDX:EAX rezultatul
- **div dword [v]** ; împărțire EDX:EAX la v

3.3. DIRECTIVE

Directivele indică modul în care sunt generate codul și datele în momentul asamblării.

3.3.1.1. Directiva SEGMENT

Directiva SEGMENT permite direcționarea octeților de cod sau date emiși de către un asamblor înspre segmentul precizat, segment care poartă un nume și are asociate diverse caracteristici.

SEGMENT *nume* [*tip*] [*ALIGN=alinie*] [*combinare*] [*utilizare*] [*CLASS=clasă*]

Numelui segmentului i se asociază ca valoare adresa de segment (32 biți) corespunzătoare poziției segmentului în memorie în faza de execuție. În acest sens, asamblorul NASM pune la dispoziție și simbolul special \$\$ care este echivalent cu adresa segmentului curent, acesta având însă avantajul că poate fi utilizat în orice context, fără a fi necesar să fie cunoscut numele segmentului în care ne aflăm.

Cu excepția numelui, toate celelalte câmpuri sunt opționale atât din punct de vedere a prezenței cât și a ordinii în care sunt specificate.

Argumentele opționale *tip*, *alinie*, *combinare*, *utilizare* și '*clasa*' dau editorului de legături și asamblorului indicații referitoare la modul de încărcare și atributele segmentelor.

Tip permite selectarea unui model de folosire al segmentului, având la dispoziție următoarele opțiuni:

- **code** (sau **text**) - segmentul va conține cod, conținutul nu poate fi scris dar se poate citi sau executa

- **data** (sau **bss**) - segment de date permițând citire și scriere însă nu și execuție (valoare implicită)
- **rdata** - segment din care se poate doar citi, menit a conține definiții de date constante

Argumentul opțional *alinie* specifică multiplul numărului de octeți la care trebuie să înceapă segmentul respectiv. Alinierile acceptate sunt puteri a lui 2, între 1 și 4096.

Dacă argumentul *alinie* lipsește, atunci se consideră implicit că este vorba despre o aliniere ALIGN=1, adică segmentul poate începe la orice adresă.

Argumentul opțional *combinare* controlează modul în care segmente cu același nume din cadrul altor module vor fi combinate cu segmentul în cauză la momentul editării de legături. Valorile posibile sunt:

- **PUBLIC** - indică editorului de legături să concateneze acest segment cu alte eventuale segmente cu același nume, obținându-se un unic segment a cărei lungime este suma lungimilor segmentelor componente.
- **COMMON** - specifică faptul că începutul acestui segment trebuie să se suprapună peste începutul tuturor segmentelor ce au același nume. Se obține un segment având dimensiunea egală cu cea a celui mai mare segment având același nume.
- **PRIVATE** - indică editorului de legături că acest segment nu este permis a fi combinat cu altele care poartă același nume.
- **STACK** - segmentele cu același nume vor fi concatenate. În faza de execuție segmentul rezultat va fi segmentul stivă.

Implicit, dacă nu se specifică o metodă de combinare, orice segment este considerat PUBLIC.

Argumentul *utilizare* permite optarea pentru altă dimensiune de cuvânt decât cea de 16 biți, care este implicită în lipsa precizării acestui argument.

Argumentul '*clasa*' are rolul de a permite stabilirea ordinii în care editorul de legături plasează segmentele în memorie. Toate segmentele având aceeași clasă vor fi plasate într-un bloc contiguu de memorie indiferent de

ordinea lor în cadrul codului sursă. Nu există o valoare implicită de inițializare pentru acest argument, el fiind nedefinit în lipsa specificării, ducând în consecință la evitarea concatenării într-un bloc continuu a tuturor segmentelor definite astfel.

segment code use32 class=CODE segment data use32 class=DATA

3.3.2. Directive pentru definirea datelor

definire date = declarare (specificarea atributelor) + *alocare* (rezervarea sp. de memorie necesar).

(UNICA !!!) (NU e unica !!!) (UNICA !!!!)

In C – 17 module (fișiere separate) ; A1- variabilă globală (int A1 + 16 declarații de data extern int A1)
LINKER-ul este responsabil pt verificarea DEPENDENTELOR dintre module !

Structura unei Variable = ([nume], set_de_atribute, [adresa/referință, valoare])

Variabilele dinamice NU AU NUME !!!!!

P=new(...); p=malloc(...); ...free... (Diferența between POINTER and DYNAMIC variables !!!!)

(Nume, set_de_atribute) = parametrii formali ai funcțiilor și procedurilor !!!

Set_de_atribute = (TD, Domeniu_de_vizibilitate (scope), durata de viață (lifetime/extent), clasa de memorie)
Memory class (in C) = (auto, register, static, extern)

tipul de dată = dimensiunea de reprezentare – octet, cuvânt, dublucuvânt, quadword

Forma generală a unei linii sursă în cazul unei declarații de date este:

[*nume*] *tip_data lista_expresii* [*;comentariu*]

sau [*nume*] *tip_alocare factor* [*;comentariu*]

sau [*nume*] **TIMES** *factor tip_data lista_expresii* [*;comentariu*]

unde *nume* este o etichetă prin care va fi referită data. Tipul rezultă din tipul datei (dimensiunea de reprezentare) iar valoarea este adresa la care se va găsi în memorie primul octet rezervat pentru data etichetată cu numele respectiv.

factor este un număr care indică de câte ori se repetă lista de expresii care urmează în paranteză.

Tip_data este o directivă de definire a datelor, una din următoarele:

DB - date de tip octet (BYTE)

DW - date de tip cuvânt (WORD)

DD - date de tip dublucuvânt (DWORD)

DQ - date de tip 8 octeți (QWORD - 64 biți)

DT - date de tip 10 octeți (TWORD - utilizate pentru memorarea constantelor BCD sau constantelor reale de precizie extinsă)

De exemplu, secvența următoare definește și inițializează cinci variabile de memorie:

```
segment data
    var1 DB 'd' ;1 octet
        .a DW 101b ;2 octeți
    var2 DD 2bfh ;4 octeți
        .a DQ 307o ;8 octeți (1 quadword)
        .b DT 100 ;10 octeți
```

Variabilele `var1` și `var2` sunt definite folosind etichete obișnuite, cu vizibilitate la nivelul întregului cod sursă, în timp ce `.a` și `.b` sunt etichete locale, accesul la aceste variabile impunând următoarele constrângeri:

- acestea se pot accesa cu numele local, adică `.a` sau `.b`, până în momentul definirii unei alte etichete obișnuite (ele fiind locale etichetei ce le preced);
- pot fi accesate de oriunde prin numele lor complet: `var1.a`, `var2.a` sau `var2.b`.

Valoarea de inițializare poate fi și o expresie, ca de exemplu în

```
var1 DW (1002/4+1)
```

După o directivă de definire a datelor pot să apară mai multe valori, permițându-se astfel declararea și inițializarea de tablouri. De exemplu, declarația

```
Tablou DW 1,2,3,4,5
```

crează un tablou de 5 întregi reprezentați pe cuvinte având valorile respectiv 1,2,3,4,5. Dacă valorile de după directivă nu încap pe o singură linie se pot adăuga oricâte linii este necesar, linii ce vor conține numai directiva și valorile dorite. Exemplu:

```
Tabpatrate DD 0, 1, 4, 9, 16, 25, 36
             DD 49, 64, 81
             DD 100, 121, 144, 169
```

Tip_alocare este o directivă de rezervare de date neinițializate:

RESB - date de tip octet (BYTE)

RESW - date de tip cuvânt (WORD)

RESD - date de tip dublucuvânt (DWORD)

RESQ - date de tip 8 octeți (QWORD - 64 biți)

REST - date de tip 10 octeți (TWORD – 80 biți)

factor este un număr care indică de câte ori se repetă tipul alocării precizate

De exemplu

```
Tabzero RESW 100h    rezervă 256 de cuvinte pentru tabloul Tabzero
```

NASM does not support the MASM/TASM syntax of reserving uninitialised space by writing DW ? or similar things: this is what it does instead. The operand to a RESB-type pseudo-instruction is a **critical expression** (toți operanzii care intervin în calcul trebuie să fie cunoscuți în momentul în care expresia este întâlnită). Ex:

```
buffer:      resb   64    ; reserve 64 bytes
wordvar:     resw    1     ; reserve a word
realarray    resq   10     ; array of ten reals
```

Directiva TIMES permite asamblarea repetată a unei instrucțiuni sau definiții de dată:

TIMES *factor tip_data expresie*

De exemplu Tabchar TIMES 80 DB 'a'

crează un tablou de 80 de octeți inițializați fiecare cu codul ASCII al caracterului 'a'.

matrice10x10 times 10*10 dd 0

va furniza 100 de dublucuvinte dispuse continuu în memorie începând de la adresa asociată etichetei matrice10x10.

TIMES can also be applied to instructions:

TIMES *factor instrucțiune*

TIMES 32 add eax, edx ; having as effect $EAX = EAX + 32 * EDX$

3.3.3. Directiva EQU

Directiva EQU permite atribuirea, în faza de asamblare, unei valori numerice sau șir de caractere unei etichete fără alocarea de spațiu de memorie sau generare de octeți. Sintaxa directivei EQU este

nume EQU expresie

Exemple:

END_OF_DATA	EQU '!'
BUFFER_SIZE	EQU 1000h
INDEX_START	EQU (1000/4 + 2)
VAR_CICLARE	EQU i

Prin utilizarea de astfel de echivalări textul sursă poate deveni mai lizibil. Se observă asemănarea etichetelor echivalate prin directiva EQU cu constantele din limbajele de programare de nivel înalt.

Expresia pentru echivalarea unei etichete definite prin directiva EQU poate conține la rândul ei etichete definite prin EQU:

TABLE_OFFSET	EQU 1000h
INDEX_START	EQU (TABLE_OFFSET + 2)
DICTIONAR_STAR	EQU (TABLE_OFFSET + 100h)